


# Simulation of the signal integrity of the ELCO B 25 connectors

This document is a guidance for the use of the SPICE simulation models of the ELCO B 25 connector family. In a first step, the assignment between the signals on the printed circuit boards going through the plugged connector and the SPICE simulation model is shown. In the following steps, an exemplary SPICE input deck is created and will be explained and the simulation results are given at the end of this document.

## 1 Introduction

The following figure shows two printed circuit boards which are connected by an ELCO B 25 7-Row 2mm connector pair. The top view to both PCBs is given in the left part of the figure, while the right one shows the side view.



**Fig. 1** Top and side view of the printed circuit boards

We will look at a certain part of the connector, containing 6 signals. It is assumed that the signals which are marked with blue circles are differentially driven by line drivers with rise times of  $t_{\text{rise}} = 500\text{ps}$ . The differential signal line which is marked with a red circle leads to a receiver. Now the goal of the simulation is, to analyze the crosstalk to this differential receiver line pair when the other 2 line pairs change simultaneous from 0 to  $\pm 500\text{mV}$ .

## 2 Transferring the pin description to a SPICE input deck

A main issue in setting up a simulation run is the correct use of the SPICE models of the ELCO B 25 connectors. In this case, it is assumed, that connectors with shields on the top and bottom sides are used (e.g. part no. 17-7200-175-100-001 for the male connector and part no. 27-7200-125-103-001 for the female connector).

A very important point which has to be considered is, that the SPICE simulation models represent only a reduced part of whole connector. While the physical connector as shown in the sketch above consists of 5 pin rows (a-e) and 25 pin columns (1-25) plus two rows (f and z) for the shields on the top and bottom of the connector, the simulation model is restricted to the same number of pin rows (a-e) and 7 pin columns (1-7) plus the shields (row f and z). The reason for this restriction is the circuit size of the model. Creating a model for the whole connector would slow down the SPICE simulator. However, for the most applications, a subset of the whole connector is sufficient to simulate a certain signal configuration.




Fig. 2 Pin array of the simulation model

The following two examples describe the two possible application scenarios:

### Simulation of a pin matrix inside a connector

As shown below the usable pin matrix is surrounded by pin columns at the left and the right side. Therefore, the most left (col. 1) and the most right (col. 7) pin column in the model must be set to a defined potential (here GROUND) and the usable pin matrix is restricted to column 2 to 6. This restriction is necessary to define the correct typical impedance of the usable pin array, if this is surrounded by other pins. That means, the pin matrix defined by pins no. (a2 .. e6) can be used for applying active signals while pin columns (a1.. e1) and (a7 .. e7) must be set to a defined state to get correct boundary conditions.




Fig. 3 Usable pin array inside a connector

### Simulation of a pin matrix at the end of a connector




Fig. 4 Usable pin array at the end of a connector

In this case the usable pin matrix is located at the left end of the connector. Therefore, only the most right (col. 7) pin column in the model must be set to a defined potential (here GROUND) and the usable pin matrix is restricted to column 1 to 6.


### 3 Pin assignment

For this application the library file " elco\_2mm\_xtalk\_shield.lib" is selected which represents the used connector pair. In the header of the library file, we can find the following explanation how to use this SPICE model:

Pin assignment

The connection nodes of the model are named as follows:

- the first block of pins is assigned to the simulated signals of the receptacle contact tails
- the second block of pins is assigned to the simulated signals of the vertical header contact tails



## Usage in a SPICE simulation circuit

-----

The connector subcircuit call statement syntax is:

```
x_connector_elco_2mm_xtalk_shield
+ f_f
+ f_a1  f_a2  f_a3  f_a4  f_a5  f_a6  f_a7
+ f_b1  f_b2  f_b3  f_b4  f_b5  f_b6  f_b7
+ f_c1  f_c2  f_c3  f_c4  f_c5  f_c6  f_c7
+ f_d1  f_d2  f_d3  f_d4  f_d5  f_d6  f_d7
+ f_e1  f_e2  f_e3  f_e4  f_e5  f_e6  f_e7
+ f_z
+
+ m_f
+ m_a1  m_a2  m_a3  m_a4  m_a5  m_a6  m_a7
+ m_b1  m_b2  m_b3  m_b4  m_b5  m_b6  m_b7
+ m_c1  m_c2  m_c3  m_c4  m_c5  m_c6  m_c7
+ m_d1  m_d2  m_d3  m_d4  m_d5  m_d6  m_d7
+ m_e1  m_e2  m_e3  m_e4  m_e5  m_e6  m_e7
+ m_z
+
+ elco_2mm_xtalk_shield
```

Replace the symbolic node names with the real node names in your circuit in the same order !!

## Notes to pin assignment:

-----

- f\_a1, f\_a2, ..., m\_e7 are signal pins.
- f\_f, f\_z ,m\_f and m\_z are pins of the shielding plates around the signal pins.

Due to the SPICE syntax, the nodes of a sub-circuit must be written in a straight forward order. The description above corresponds to the description in **Fig. 2**. The actual pin configuration of the example in Fig. 1 will now be converted into an equivalent SPICE description step by step.


## 1. sub-circuit name

The invocation of the B 25 SPICE model starts with the circuit element for a subcircuit (X)

**X\_connector**

## 2. pin assignment

According to the description in Fig. 5, the pin assignment starts with the input to the female connector. In this case, it is assumed that all pins except the signal pins are set to the absolute ground potential "0". Please see Fig. 5 for details about the numbering of the columns of the connector. As mentioned earlier, the model does not represent all 25 columns of the connector. Therefore, the column numbers of the pins on the PCB must be transferred to appropriate column numbers in the model. In the example above, the original pin **a8** on the PCB will become pin **a1** in the SPICE model.



**Fig. 5 Local coordinate system for the column numbers of the SPICE model**

The first row in the model (z) is the connection of the upper shield to ground and is set to 0.

+ 0

Please note that SPICE allows to truncate an input line and to continue in the next line. The '+' sign must be placed as the first character in this new line. The following rows a and b contains three differential signal pairs where pair Sa2f/Sb2f and pair Sa6f/Sb6f are aggressor pairs and Sa4f/Sb4f is the victim pair:

```
+ 0    Sa2f  0    Sa4f  0    Sa6f  0
+ 0    Sb2f  0    Sb4f  0    Sb6f  0
```

The following 3 lines describes the input for row c through e of the female connector.

```
+ 0    0    0    0    0    0
+ 0    0    0    0    0    0
+ 0    0    0    0    0    0
```

The last row in the model (f) is the connection to the lower shield to ground and is therefore set to 0.

+ 0

Now the following 7 lines define the inputs to the male connector starting again with row z for the shield and continue with row a through f.

```
+ 0
+ 0    Sa2m  0    Sa4m  0    Sa6m  0
+ 0    Sb2m  0    Sb4m  0    Sb6m  0
+ 0    0    0    0    0    0
+ 0    0    0    0    0    0
+ 0    0    0    0    0    0
+ 0
```

The final line is the call of the sub-circuit:

```
+ elco_2mm_xtalk_shield
```

The transmission lines to the connectors are assumed to be lossless lines without any crosstalk with a characteristic impedance of  $50\Omega$ . The drivers has a risetime of  $t_{rise} = 500ps$  and an internal resistance of  $50\Omega$ .

The overall spice netlist can then be composed as shown below. This example is given for SPICE3f5 from University of California at Berkeley, which can be assumed to be a standard in SPICE notation:

```

*****
* Simulation of the crosstalk behavior of the ELCO B25 connector *
* if 2 differential signal pairs switches simultaneously from 0 to +/-0.5V. *
*****

* Library of shielded ELCO B25 2mm connector
.include elco_2mm_xtalk_shield.lib

* ---- Aggressor differential transmission lines to the female connector ----
xline1 101 Sa2f 201 Sb2f 0 pulse
xline2 102 Sa6f 202 Sb6f 0 pulse

* ---- Aggressor differential transmission lines to the male connector ----
xload1 Sa2m 401 Sb2m 501 0 load
xload2 Sa6m 402 Sb6m 502 0 load

* ---- Victim differential transmission lines to the connector ----
xline3 103 Sa4f 203 Sb4f 0 load
xload3 Sa4m 403 Sb4m 503 0 load

* ---- ELCO B25 Connector ----
X_connector
*** female ***
+ 0
+ 0 Sa2f 0 Sa4f 0 Sa6f 0
+ 0 Sb2f 0 Sb4f 0 Sb6f 0
+ 0 0 0 0 0 0 0
+ 0 0 0 0 0 0 0
+ 0 0 0 0 0 0 0
+ 0
*** male ***
+ 0
+ 0 Sa2m 0 Sa4m 0 Sa6m 0
+ 0 Sb2m 0 Sb4m 0 Sb6m 0
+ 0 0 0 0 0 0 0
+ 0 0 0 0 0 0 0
+ 0 0 0 0 0 0 0
+ 0
+ elco_2mm_xtalk_shield

* ---- Subcircuit of differ. transmission line with pulse input of 50 ohm source
.subckt pulse 2 3 5 6 99
vin1 1 99 pulse 0 1V 0 500ps 150ps 4ns 8ns
rin1 1 2 50
T1 2 99 3 99 Z0=50 TD=1n
vin2 4 99 pulse 0 -1V 0 500ps 150ps 4ns 8ns
rin2 4 5 50
T2 5 99 6 99 Z0=50 TD=1n
.ends

* ---- Subcircuit of differential transmission line with 50 ohm termination ----
.subckt load 1 2 3 4 99
T1 1 99 2 99 Z0=50 TD=1n
rload1 2 0 50
T2 3 99 4 99 Z0=50 TD=1n
rload2 4 0 50
.ends

* ---- Transient analysis ----
.tran 50p 4n 2n

.end

```

## 4 Results of the SPICE simulation

The following diagrams show some results from this simulation run. The most important information is given in Fig. 6, where the differential crosstalk is shown. The voltage S4f shows the near end crosstalk at the input to the female connector while S4m shows the far end crosstalk voltage at the input to the male connector.




Fig. 6 Crosstalk on the 'victim' line in signal S4m and S4f

The following diagram shows the signal integrity of an active signal, here S2f on its way through the B 25 connector. The back reflection and the transmission is shown in Fig. 7.




Fig. 7 Reflection S2f and transmission S2m for one aggressor line



## 5 Using example

This example consists of the following 4 files:

run_simulation.bat	(batch file to run simulation)
sim_xtalk_shield.cir	(Circuit file for simulation)
plot_xtalk_shield.cir	(Circuit file for plot)
elco_2mm_xtalk_shield.lib	(Spice model)

- Before simulation you have to install SPICE on your computer. Public available SPICE3f5 versions can be download from the following sources:  
[www.uni-duisburg.de/FB9/EBS/hauptteil\\_software\\_en.html](http://www.uni-duisburg.de/FB9/EBS/hauptteil_software_en.html) (in English language)  
[www.uni-duisburg.de/FB9/EBS/hauptteil\\_software.html](http://www.uni-duisburg.de/FB9/EBS/hauptteil_software.html) (same in German language)  
[www.uni-duisburg.de/FB9/EBS/spice3f5.exe](http://www.uni-duisburg.de/FB9/EBS/spice3f5.exe) (or direct download link)

Instead you can also download Winspice (different program) from:

[www.winspice.com](http://www.winspice.com)

- Then you have to edit the run\_simulation.bat file, e.g. with a text editor.  
You have to edit the path to your SPICE simulator, marked red below.

run\_simulation.bat:

(example with Winspice)

```
c:\programme\winspice\wspice3 -b -r sim_xtalk_shield.out sim_xtalk_shield.cir  
c:\programme\winspice\wspice3 plot_xtalk_shield.cir
```

(example with SPICE3f5)

```
C:\Program Files\Spice_Win\bin\spice3 -b -r sim_xtalk_shield.out sim_xtalk_shield.cir  
C:\Program Files\Spice_Win\bin\spice3 plot_xtalk_shield.cir
```

- Now you can start the simulation by running the run\_simulation.bat  
When finished the two plots defined in plot\_xtalk\_shield.cir will be shown.